AD-A131 936   A COMPUTER SUBROUTINE FOR EVALUATING POLYGAMMA FUNCTIONS FOR COMPLEX ARGU..(U) ARMY ARMAMENT RESEARCH AND DEVELOPMENT COMMAND ABERDEEN PROVI.. J N WALBERT   1/1

UNCLASSIFIED   JUL 83 ARBRL-TR-02510 SBI-AD-F300 298   F/G 9/2   NL

END
DATE
FILMED
DTIC

1.0 1.1 1.25 1.4 1.6 1.8 2.0 2.2 2.5 2.8 3.2 3.6 4.0 4.5 5.0 5.6

MICROCOPY RESOLUTION TEST CHART
NATIONAL BUREAU OF STANDARDS-1963-A

TECHNICAL REPORT ARBRL-TR-02510

A COMPUTER SUBROUTINE FOR EVALUATING POLYGAMMA FUNCTIONS FOR COMPLEX ARGUMENTS

James N. Walbert

July 1983

US ARMY ARMAMENT RESEARCH AND DEVELOPMENT COMMAND
BALLISTIC RESEARCH LABORATORY
ABERDEEN PROVING GROUND, MARYLAND

UNCLASSIFIED

SECURITY CLASSIFICATION OF THIS PAGE (When Data Entered)

REPORT DOCUMENTATION PAGE

READ INSTRUCTIONS BEFORE COMPLETING FORM

1. REPORT NUMBER TECHNICAL REPORT ARBRL-TR-02510	2. GOVT ACCESSION NO. AD-A131936	3. RECIPIENT'S CATALOG NUMBER
4. TITLE (and Subtitle) A COMPUTER SUBROUTINE FOR EVALUATING POLYGAMMA FUNCTIONS FOR COMPLEX ARGUMENTS		5. TYPE OF REPORT & PERIOD COVERED Final Report 6. PERFORMING ORG. REPORT NUMBER
7. AUTHOR(s) JAMES N. WALBERT		8. CONTRACT OR GRANT NUMBER(s)
9. PERFORMING ORGANIZATION NAME AND ADDRESS US Army Ballistic Research Laboratory ATTN: DRDAR-BLI Aberdeen Proving Ground, MD 21005		10. PROGRAM ELEMENT, PROJECT, TASK AREA & WORK UNIT NUMBERS 1L162618AH80
11. CONTROLLING OFFICE NAME AND ADDRESS US Army Armament Research & Development Command US Army Ballistic Research Laboratory (DRDAR-BLA-S) Aberdeen Proving Ground, MD 21005		12. REPORT DATE July 1983 13. NUMBER OF PAGES 25
14. MONITORING AGENCY NAME & ADDRESS(if different from Controlling Office)		15. SECURITY CLASS. (of this report) UNCLASSIFIED 15a. DECLASSIFICATION/DOWNGRADING SCHEDULE

16. DISTRIBUTION STATEMENT (of this Report)

Approved for public release; distribution unlimited.

17. DISTRIBUTION STATEMENT (of the abstract entered in Block 20, if different from Report)

18. SUPPLEMENTARY NOTES

19. KEY WORDS (Continue on reverse side if necessary and identify by block number)

Psi function
Gamma function
Polygamma functions
Asymptotic series
Recursion
Digamma function

20. ABSTRACT (Continue on reverse side if necessary and identify by block number) jmk

The series and formulas used to evaluate polygamma functions are discussed in sufficient detail to enable the user to correctly implement the computer code. Estimates of precision are given, and a listing of the code appears in the appendix.

DD FORM 1 JAN 73 1473 EDITION OF 1 NOV 65 IS OBSOLETE

UNCLASSIFIED

SECURITY CLASSIFICATION OF THIS PAGE (When Data Entered)

TABLE OF CONTENTS

NOTATION

This user's manual is designed to assist the mathematician or programmer using the BRL Polygamma Function subroutine. FORTRAN symbols for variables and arithmetic operations are used in the body of the report for consistency with excerpts from the coding.

As an aid to the reader unfamiliar with standard FORTRAN, the following symbols are defined:

	Symbol	Operation	Algebraic Notation	FORTRAN Notation
1.	+	add	a + b	= A + B
2.	-	subtract	a − b	= A − B
3.	*	multiply	a × b	= A * B
4.	/	divide	a ÷ b	= A / B

Numbers are written in specific ways to define their type:

1. Integer: 2
2. Real: 2. or 2.0
3. Standard notation 2.78×10^5: 2.78 E+05

   (double precision) 2.78 D+05

I. INTRODUCTION

The polygamma functions appear in numerous mathematical expressions related to the evaluation of stresses in tapered sections of gun tubes. The Ballistic Research Laboratory (BRL) has need of special function subroutines such as the one described in this report for use in large numerical analysis computer codes for stress simulations and experimental data evaluation. There are some specific requirements placed on such subroutines. In particular, memory requirements should be minimal, since part of the code may be run on small computers in the pre- or post-processing of the data. For this same reason, the subroutine should be as machine independent as possible. That is, it should contain no "special tricks" related to a specific machine. Finally, the code must achieve the greatest precision possible, since the output is merely a component of other computations.

The polygamma function subroutine described in this report satisfies these conditions over a broad range of application requirements. It is designed for complex arguments in double-precision; the subroutine is written in FORTRAN IV and does not require implicit complex arithmetic. Examples run on the CDC 7600 computer at BRL have provided a precision of 20-22 significant digits. Examples run on an HP-1000 minicomputer at BRL have provided a precision of 13-15 significant digits.

II. INPUT AND OUTPUT VARIABLES

The subroutine statement is

SUBROUTINE POLYG(X, Y, N, POLYR,POLYI,IERR).

The input variables are X, Y, and N. X and Y are the real and imaginary parts, respectively, of the complex argument Z=X+iY at which the function is to be evaluated. X and Y are both double-precision real variables. N is an integer variable which specifies which polygamma function is to be evaluated. Specifically,

N = 0 specifies computation of the Psi function;

N = 1,2,3,... specifies computation of the derivative of the Psi function of order 1, 2, 3,..., respectively.

NOTE: If Z=X+iY is in the left half plane, then N must be less than 5.

The output variables are POLYR,POLYI, and IERR. POLYR and POLYI are double-precision real variables which are the real and imaginary parts, respectively, of the requested function. IERR is an integer variable specifying errors detected by the subroutine. Specifically, if

IERR = 0, no errors occurred;

= 1, negative N was requested;

= 2, input argument X+iY was zero or a negative integer;

= 3, N>4 and X+iY was in the left half plane.

Appropriate error messages are printed by the subroutine to accompany the nonzero values of IERR.

III. METHOD OF COMPUTATION

A. Definitions. The gamma function, $\Gamma(z)$, is defined by*

$$\Gamma(Z) = \lim_{n\to\infty} \frac{n!\,n^Z}{Z(Z+1)\ \dots\ (Z+n)} , \qquad (1)$$

$Z \neq 0, -1, -2, \dots$

The Psi function, $\psi(Z)$, is defined by

$$\psi(Z) = \Gamma'(Z)/\Gamma(Z). \qquad (2)$$

$\psi(Z)$ is the polygamma function of order zero. In general, the nth-order polygamma function $\psi^{(n)}(Z)$ is defined by

$$\psi^{(n)}(Z) = \frac{d^n}{dZ^n}(\psi(Z)) , \qquad (3)$$

where n is a positive integer.

The method of computation used by the subroutine is evaluation of the asymptotic series for $\psi^{(n)}(Z)$. If the magnitude of Z is small, recursion formulas are used.

B. Asymptotic Series. For values of Z not on the negative real axis,

$$\psi^{(n)}(Z) \sim (-1)^{n-1}\Big[\frac{(n-1)!}{Z^n} + \frac{n!}{2Z^{n+1}} + \sum_{k=1}^{\infty} B_{2k} \frac{(2k+n-1)!}{(2k)!\,Z^{2k+n}}\Big] , \qquad (4)$$

*All of the formulas used in this report may be found in the Handbook of Mathematical Functions of the National Bureau of Standards.

as $Z\to\infty$, where the symbol ! denotes the factorial operator, and the B_{2k} are the Bernoulli numbers, defined later in this section. In the case n = 0,

$$\psi(Z) \sim \ln Z - \frac{1}{2Z} - \sum_{n=1}^{\infty} \frac{B_{2n}}{2nZ^{2n}} . \qquad (5)$$

C. Recurrence and Reflection Formulas. Through trial computer runs it was found that the asymptotic series was most effective for values of Z such that $|Z|>10$. When $|Z|<10$ the subroutine uses the recurrence formula

$$\psi^{(n)}(Z+1) = \psi^{(n)}(Z) + (-1)^n n!\ Z^{-n-1} \qquad (6)$$

When Re $\{Z\}$ is negative, and Z is not an integer, the values of $\psi^{(n)}(Z)$ are defined in terms of $\psi^{(n)}(1-Z)$ by use of the reflection formula

$$\psi^{(n)}(Z)=(-1)^n\psi^{(n)}(1-Z) - \pi \frac{d^n}{dZ^n} \cot (\pi Z). \qquad (7)$$

It was decided that the analytic expression for the first four derivatives of the cotangent function would be programmed directly, hence the restriction on N.

D. Programming Methods. The line numbers given in this section refer to the program listing in Appendix A. For operation on various computers, the value of PI in line 11 can be increased or decreased in precision. It should be as precise as a particular computer will allow. Lines 12-30 define the Bernoulli numbers B_2, B_4,...B_{38} for use in the asymptotic series. These numbers are expressed as quotients, with exact numerators and denominators. This allows them to be computed to the full precision of the computer. Clearly, then, no more than 38 terms of the asymptotic series are used. In fact, in all ranges of values of the argument used for testing, 26 terms was the maximum ever required to achieve optimal precision. Line 31 sets this maximum number of terms to 38.

Line 32 sets the cutoff value below which recurrence must be used. Lines 33-50 check for input errors. If $|Z|$ is within 1.D-30 of zero or a negative integer, an error code IERR=2 is set. This value (line 33) may be changed to suit a particular computer's range. Lines 41-46 convert Z from the left half plane to the right half plane, if necessary. Lines 47-50 convert Z to the first quadrant, if necessary, to simplify the computations. The sign of the imaginary part of Z is saved in the variable SGNI for use in line 58.

Evaluation of the appropriate function is accomplished in lines 51-67. If $|Z|$ is less than the cutoff value set in line 32, then SUBROUTINE RECUR is called to implement the recurrence Formula (6). If N>0, SUBROUTINE PGAM is called to evaluate the polygamma function. If N=0, SUBROUTINE PSI is called to evaluate the Psi function. In lines 57 and 58, the real (ADDR) and imaginary (ADDI) parts, if any, due to recurrence are added to the function values POLYR and POLYI, respectively. In line 58, the sign of the imaginary part is corrected for the quadrant in which Z lies. Lines 60-67 handle the reflection from the left half plane, using SUBROUTINE COTAN to evaluate the appropriate derivative of the cotangent function.

Each of the subroutines will be discussed in detail. SUBROUTINE RECUR (lines 86-111) implements Formula (6). For N>0, lines 94-97 compute N!. Lines 98-106 sum the powers of Z computed in SUBROUTINE CPOWR (lines 193-206). In lines 108 and 109, the sums of the real and imaginary parts of the powers of Z, ADDR and ADDI, respectively, are multiplied by N! and SGN, which is $(-1)^n$.

SUBROUTINE PGAM (lines 112-165) implements Formula (4) N>0. Lines 119-123 compute N! and (N-1)!. Lines 124-130 compute Z^{-N} and $\frac{1}{2}\ Z^{-N-1}$, multiply them by (N-1)! and N!, respectively, and sum these terms. The remainder of the subroutine sums the terms of the series, ending with NTERMS or when the magnitude of the individual terms stops decreasing.

SUBROUTINE PSI (lines 166-192) implements Formula (5). It uses the same technique for summing the asymptotic series as was used in SUBROUTINE PGAM. SUBROUTINE COTAN (lines 207-256) evaluate the appropriate derivative of the cotangent function, according to the formulas

$$\frac{d}{dz} \cot(\pi z) = -\pi(1 + \cot^2(\pi z)) \quad , \tag{8}$$

$$\frac{d^2}{dz^2} \cot(\pi z) = 2\pi^2 (\cot(\pi z) + \cot^3(\pi z)) \quad , \tag{9}$$

$$\frac{d^3}{dz^3}\cot(\pi z) = -2\pi^3(1 + 4\cot^2(\pi z) + 3\cot^4(\pi z)) \quad , \tag{10}$$

$$\frac{d^4}{dz^4}\cot(\pi z) = 8\pi^4(2\cot(\pi z) + 5\cot^3(\pi z) + 3\cot^5(\pi z)) \quad , \tag{11}$$

where

$$\cot(\pi z) = \frac{4\sin(\pi x)\cos(\pi x)}{e^{2\pi y}+e^{-2\pi y}+4\sin^2\pi x-2} + i\,\frac{(e^{-2\pi y}-e^{2\pi y})}{e^{2\pi y}+e^{-2\pi y}+4\sin^2\pi x-2} \quad , \tag{12}$$

for $z = x+iy$.

IV. CONCLUSIONS

The subroutine described in this report provides values of the polygamma functions for complex argument. Although precision will vary from one computer to another, the subroutine has achieved 22-digit precision on the CDC 7600.

ACKNOWLEDGMENTS

The author is indebted to Mr. A.S. Elder for his suggestion concerning the recursion formulas, and to Mr. Erik Benck for his assistance in evaluating the precision of the subroutine.

APPENDIX A

LISTING OF SUBROUTINE POLYG

```
FTN4,L
      SUBROUTINE POLYG(X,Y,N,POLYR,POLYI,IERR)
C
C  ASYMPTOTIC SERIES SUBROUTINE FOR
C    THE POLYGAMMA FUNCTION
C
C     J.WALBERT,   OCT, 1981
C
      IMPLICIT DOUBLE PRECISION (A-H,O-Z)
      DIMENSION B(52)
      PI=3.14159265358979323840D0
      B(2)=1.D0/6.D0
      B(4)=-1.D0/30.D0
      B(6)=1.D0/42.D0
      B(8)=-1.D0/30.D0
      B(10)=5.D0/66.D0
      B(12)=-691.D0/2730.D0
      B(14)=7.D0/6.D0
      B(16)=-3617.D0/510.D0
      B(18)=43867.D0/798.D0
      B(20)=-174611.D0/330.D0
      B(22)=854513.D0/138.D0
      B(24)=-236364091.D0/2730.D0
      B(26)=8553103.D0/6.D0
      B(28)=-23749461029.D0/870.D0
      B(30)=8615841276005.D0/14322.D0
      B(32)=-7709321041217.D0/510.D0
      B(34)=2577687858367.D0/6.D0
      B(36)=-26315271553053477373.D0/1919190.D0
      B(38)=2929993913841559.D0/6.D0
      NTERMS=38
      CUTOFF=10.D0
      ZERO=1.D-30
      ZR=X
      ZI=Y
      IF(N.LT.0) GOTO 100
      IF(DABS(ZI).GT.ZERO.OR.ZR.GT.ZERO) GOTO 1
      INT=ZR
      TEST=DBLE(FLOAT(INT))
      IF(DABS(TEST-ZR).LE.ZERO) GOTO 200
1     REFLEC=-1.D0
      IF(ZR.GE.0.D0) GOTO 5
      IF(N.GT.4) GOTO 300
      REFLEC=1.D0
      ZR=1.D0-ZR
      ZI=-ZI
5     SGNI=1.D0
      IF(ZI.GE.0.D0) GOTO 10
      SGNI=-1.D0
      ZI=DABS(ZI)
10    ADDR=0.D0
      ADDI=0.D0
      W=DSQRT(ZR*ZR+ZI*ZI)
      IF(W.LT.CUTOFF) CALL RECUR(W,ZR,ZI,N,CUTOFF,ADDR,ADDI)
```

```
      IF(N.GT.0) CALL PGAM(ZR,ZI,N,NTERMS,B,POLYR,POLYI)
      IF(N.EQ.0) CALL PSI(ZR,ZI,NTERMS,B,POLYR,POLYI)
      POLYR=POLYR+ADDR
      POLYI=SGNI*(POLYI+ADDI)
      IF(REFLEC.LT.0.D0) GOTO 999
      IF(MOD(N,2).EQ.0) GOTO 15
      POLYR=-POLYR
      POLYI=-POLYI
15    ZR=X
      ZI=Y
      CALL COTAN(ZR,ZI,N,PI,COTDR,COTDI)
      POLYR=POLYR-COTDR
      POLYI=POLYI-COTDI
      GOTO 999
100   IERR=1
      WRITE(1,1000)
      GOTO 999
200   IERR=2
      WRITE(1,2000)
      GOTO 999
300   IERR=3
      WRITE(1,3000)
999   RETURN
1000  FORMAT('**ERROR FROM POLYGAMMA SUBROUTINE**',/,
     *       '        NEGATIVE N REQUESTED')
2000  FORMAT('        **ERROR FROM POLYGAMMA SUBROUTINE**',/,
     *       '   INPUT ARGUMENT WAS ZERO OR A NEGATIVE INTEGER')
3000  FORMAT('   **ERROR FROM POLYGAMMA SUBROUTINE**',/,
     *       'INPUT ARGUMENT WAS IN THE LEFT HALF PLANE,',/,
     *       '  WITH REQUESTED ORDER N GREATER THAN 4')
      END
      SUBROUTINE RECUR(W,ZR,ZI,N,CUTOFF,ADDR,ADDI)
      IMPLICIT DOUBLE PRECISION (A-H,O-Z)
      NTIMES=CUTOFF-W
      IF(NTIMES.LT.1) NTIMES=1
      SGN=-1.D0
      FACT=1.D0
      IF(N.LE.0) GOTO 10
      IF(MOD(N,2).NE.0) SGN=1.D0
      EN=DBLE(FLOAT(N))
      DO 5 I=1,N
      FACT=FACT*EN
5     EN=EN-1.D0
10    ZRADD=0.D0
      K=N+1
      DO 15 I=1,NTIMES
      CALL CPOWR(ZR+ZRADD,ZI,ZPR,ZPI,K)
      U=ZPR*ZPR+ZPI*ZPI
      ADDR=ADDR+ZPR/U
      ADDI=ADDI-ZPI/U
      ZRADD=DBLE(FLOAT(I))
15    CONTINUE
      ZR=ZR+ZRADD
      ADDR=ADDR*SGN*FACT
```

```
      ADDI=ADDI*SGN*FACT
      RETURN
      END
      SUBROUTINE PGAM(ZR,ZI,N,NTERMS,B,POLYR,POLYI)
      IMPLICIT DOUBLE PRECISION (A-H,O-Z)
      DIMENSION B(52)
      PROD=1.D0
      NSTOP=N-1
      EN=1.D0
      IF(NSTOP.LT.1) GOTO 25
      DO 20 I=1,NSTOP
      PROD=PROD*EN
20    EN=EN+1.D0
25    ENM1=PROD
      PROD=PROD*EN
      CALL CPOWR(ZR,ZI,ZNR,ZNI,N)
      W=ZNR*ZNR+ZNI*ZNI
      ZNRP1=ZNR*ZR-ZNI*ZI
      ZNIP1=ZNR*ZI+ZNI*ZR
      U=2.D0*(ZNRP1*ZNRP1+ZNIP1*ZNIP1)
      POLYR=PROD*ZNRP1/U+ENM1*ZNR/W
      POLYI=-PROD*ZNIP1/U-ENM1*ZNI/W
      EK=1.D0
      SUMR=0.D0
      SUMI=0.D0
      FACT=ENM1
      BASE=EN
      DFACT=1.D0
      ZRLST=1.D30
      ZILST=1.D30
      DO 100 K=2,NTERMS,2
      FACT=FACT*BASE
      BASE=BASE+1.D0
      FACT=FACT*BASE
      BASE=BASE+1.D0
      DFACT=DFACT*EK
      EK=EK+1.D0
      DFACT=DFACT*EK
      EK=EK+1.D0
      COEF=B(K)*FACT/DFACT
      CALL CPOWR(ZR,ZI,ZNR,ZNI,K+N)
      W=ZNR*ZNR+ZNI*ZNI
      ZNR=COEF*ZNR/W
      ZNI=-COEF*ZNI/W
      IF(DABS(ZNR).GT.ZRLST.OR.DABS(ZNI).GT.ZILST) GOTO 110
      ZRLST=DABS(ZNR)
      ZILST=DABS(ZNI)
      SUMR=SUMR+ZNR
      SUMI=SUMI+ZNI
100   CONTINUE
110   POLYR=POLYR+SUMR
      POLYI=POLYI+SUMI
      IF(MOD(N,2).NE.0) GOTO 200
      POLYR=-POLYR
```

```
      POLYI=-POLYI
  200 RETURN
      END
      SUBROUTINE PSI(ZR,ZI,NTERMS,B,POLYR,POLYI)
      IMPLICIT DOUBLE PRECISION (A-H,O-Z)
      DIMENSION B(52)
      W=ZR*ZR+ZI*ZI
      IF(W.LT.1.D-30) GOTO 99
      POLYR=DLOG(DSQRT(W))-.5D0*ZR/W
      POLYI=DATAN(ZI/ZR)+.5D0*ZI/W
      SUMR=0.D0
      SUMI=0.D0
      ZRLST=1.D30
      ZILST=1.D30
      DO 10 K=2,NTERMS,2
      COEF=B(K)/DBLE(FLOAT(K))
      CALL CPOWR(ZR,ZI,ZNR,ZNI,K)
      W=ZNR*ZNR+ZNI*ZNI
      ZNR=COEF*ZNR/W
      ZNI=-COEF*ZNI/W
      IF(DABS(ZNR).GT.ZRLST.OR.DABS(ZNI).GT.ZILST) GOTO 20
      ZRLST=DABS(ZNR)
      ZILST=DABS(ZNI)
      SUMR=SUMR+ZNR
      SUMI=SUMI+ZNI
   10 CONTINUE
   20 POLYR=POLYR-SUMR
      POLYI=POLYI-SUMI
   99 RETURN
      END
      SUBROUTINE CPOWR(X,Y,U,V,NTIMES)
      DOUBLE PRECISION X,Y,U,V,W,S,T
      S=X
      T=Y
      U=S
      V=T
      IF(NTIMES.LE.1) GOTO 20
      NTM1=NTIMES-1
      DO 10 I=1,NTM1
      W=S*U-T*V
      V=T*U+S*V
   10 U=W
   20 RETURN
      END
      SUBROUTINE COTAN(ZR,ZI,N,PI,COTDR,COTDI)
      IMPLICIT DOUBLE PRECISION (A-H,O-Z)
      WI=PI*ZR
      WCOS=DCOS(WI)
      WSIN=DSIN(WI)
      IF(ZI.EQ.0.D0) GOTO 1
      UINV=DEXP(PI*ZI)
      U=1.D0/UINV
      DENOM=U*U+UINV*UINV+4.D0*WSIN*WSIN-2.D0
      CTPZR=4.D0*WSIN*WCOS/DENOM
```

```
      CTPZI=(U*U-UINV*UINV)/DENOM
      GOTO 5
1     CTPZR=WCOS/WSIN
      CTPZI=0.D0
5     IF(N.EQ.0) GOTO 60
      IF(N.GT.4) GOTO 50
      GOTO (10,20,30,40),N
10    CALL CPOWR(CTPZR,CTPZI,COTDR,COTDI,2)
      COTDR=-PI*(1.D0+COTDR)
      COTDI=-PI*COTDI
      GOTO 70
20    CALL CPOWR(CTPZR,CTPZI,COTDR,COTDI,3)
      COEF=2.D0*PI*PI
      COTDR=COEF*(CTPZR+COTDR)
      COTDI=COEF*(CTPZI+COTDI)
      GOTO 70
30    CALL CPOWR(CTPZR,CTPZI,TEMPR,TEMPI,2)
      CALL CPOWR(TEMPR,TEMPI,COTDR,COTDI,2)
      COEF=-2.D0*PI*PI*PI
      COTDR=COEF*(1.D0+4.D0*TEMPR+3.D0*COTDR)
      COTDI=COEF*(4.D0*TEMPI+3.D0*COTDI)
      GOTO 70
40    CALL CPOWR(CTPZR,CTPZI,TEMPR,TEMPI,3)
      CALL CPOWR(CTPZR,CTPZI,COTDR,COTDI,2)
      U=COTDR*TEMPR-COTDI*TEMPI
      COTRI=COTDI*TEMPR+COTDR*TEMPI
      COTDR=U
      COEF=8.D0*PI*PI*PI*PI
      COTDR=COEF*(2.D0*CTPZR+5.D0*TEMPR+3.D0*COTDR)
      COTDI=COEF*(2.D0*CTPZI+5.D0*TEMPI+3.D0*COTDI)
      GOTO 70
50    COTDR=0.D0
      COTDI=0.D0
      GOTO 70
60    COTDR=CTPZR
      COTDI=CTPZI
70    COTDR=PI*COTDR
      COTDI=PI*COTDI
      RETURN
      END
      END$
```

DISTRIBUTION LIST

No. of Copies	Organization
12	Administrator Defense Technical Info Center ATTN: DTIC-DDA Cameron Station Alexandria, VA 22314
1	Director Defense Advanced Research Projects Agency 1400 Wilson Boulevard Arlington, VA 22209
1	Director Defense Nuclear Agency Arlington, VA 22209
1	Commander US Army BMD Advanced Technology Center ATTN: BMDATC-M P.O. Box 1500 Huntsville, AL 35804
1	Commander US Army Materiel Development and Readiness Command ATTN: DRCDMD-ST 5001 Eisenhower Avenue Alexandria, VA 22333
2	Commander US Army Armament Research and Development Command ATTN: DRDAR-TSS Dover, NJ 07801
1	Commander US Army Armament Research and Development Command ATTN: DRDAR-TDC Dover, NJ 07801
2	Commander US Army Armament Research and Development Command ATTN: DRDAR-SC, B. Shulman Mr. Webster Dover, NJ 07801
7	Commander US Army Armament Research and Development Command ATTN: DRDAR-LCR-R, T. Moore DRDAR-LCR, W. Williver DRDAR-LCS-D, K. Reuben DRDAR-LCA S. Bernstein DRDAR-LCN, G. Demitrack DRDAR-LCA, B. Knutulski DRDAR-LCA, T. Davidson Dover, NJ 07801
2	Commander US Army Armament Research and Development Command ATTN: DRDAR-SE Dover, NJ 07801
1	Commander US Army Armament Research and Development Command ATTN: DRDAR-FU Dover, NJ 07801
2	Commander US Army Armament Research and Development Command ATTN: DRDAR-DP Dover, NJ 07801
2	Commander US Army Armament Research and Development Command ATTN: DRDAR-QA Dover, NJ 07801
2	Commandant US Army Infantry School ATTN: ATSH-CD-CSO-OR Fort Benning, GA 31905
1	Commander US Army Armament Materiel Readiness Command ATTN: DRSAR-LEP-L Rock Island, IL 61299

DISTRIBUTION LIST

No. of Copies	Organization
1	Director US Army ARRADCOM Benet Weapons Laboratory ATTN: DRDAR-LCB-TL Watervliet, NY 12189
1	Director US Army ARRADCOM Benet Weapons Laboratory ATTN: DRDAR-LCB, T. Simkins Watervliet, NY 12189
1	Commander US Army Aviation Research and Development Command ATTN: DRDAV-E 4300 Goodfellow Blvd. St. Louis, MO 63120
1	Director US Army Research and Technology Laboratories (AVRADCOM) Ames Research Center Moffett Field, CA 94035
1	Director US Army Air Mobility Research and Development Laboratory Ames Research Center Moffett Field, CA 94035
1	Commander US Army Communications Research and Development Command ATTN: DRSEL-ATDD Fort Monmouth, NJ 07703
1	Commander US Army Electronics Research and Development Command Technical Support Activity ATTN: DELSD-L Fort Monmouth, NJ 07703
1	Commander US Army Harry Diamond Labs 2800 Powder Mill Road Adelphi, MD 20783
1	Commander US Army Missile Command ATTN: DRSMI-R Redstone Arsenal, AL 35898
1	Commander US Army Missile Command ATTN: DRSMI-YDL Redstone Arsenal, AL 35898
1	Commander US Army Mobility Equipment Research & Development Command Fort Belvoir, VA 22060
1	Commander US Army Tank Automotive Command ATTN: DRSTA-TSL Warren, MI 48090
2	Project Manager Division Air Defense Gun ATTN: DRCPM-ADG Dover, NJ 07801
3	Project Manager Cannon Artillery Weapons System ATTN: DRCPM-CAWS Dover, NJ 07801
2	Project Manager Nuclear Munitions ATTN: DRCPM-NUC Dover, NJ 07801
2	Project Manager Tank Main Armament Systems ATTN: DRCPM-TMA Dover, NJ 07801
1	Product Manager for 30mm Ammo. ATTN: DRCPM-AAH-30mm Dover, NJ 07801
2	Product Manager M110E2 Weapon System, DARCOM ATTN: DRCPM-M110E2 Rock Island, IL 61299

DISTRIBUTION LIST

No. of Copies	Organization
3	Commander US Army Research Office P.O. Box 12211 ATTN: Technical Director Engineering Division Metallurgy & Materials Division Research Triangle Park, NC 27709
2	Commander US Army Research Office ATTN: Dr. J. Chandra Dr. F. Schmiedeshoff Research Triangle Park NC 27709
4	Director US Army Mechanics and Materials Research Center ATTN: Director (3 cys) DRXMR-ATL (1 cy) Watertown, MA 02172
1	Director US Army TRADOC Systems Analysis Activity ATTN: ATAA-SL White Sands Missile Range NM 88002
1	Commander Naval Air Systems Command Washington, DC 20360
1	Commander Naval Surface Weapons Center Silver Spring, MD 20910
1	Commander Naval Sea Systems Command Washington, DC 20360
2	Commander David W. Taylor Naval Ship Research & Development Cntr Bethesda, MD 20084
1	Commander Naval Air Development Center, Johnsville Warminster, PA 18974

No. of Copies	Organization
1	Commander Naval Missile Center Point Mugu, CA 93041
1	Commander Naval Surface Weapons Center Dahlgren VA 22448
1	Commander Naval Weapons Center China Lake, CA 93555
1	Commander Naval Research Laboratory Washington, DC 20375
1	Superintendent Naval Postgraduate School ATTN: Dir of Lib Monterey, CA 93940
1	Commander Naval Ordnance Station Indian Head, MD 20640
2	AFRPL ATTN: W. Andrepont T. Park Edwards AFB, CA 93523
1	AFATL Eglin AFB, FL 32542
1	AFWL Kirkland AFB, NM 87117
1	Director Lawrence Livermore Laboratory P.O. Box 808 Livermore, CA 94550
1	Director Los Alamos Scientific Lab P.O. Box 1663 Los Alamos, NM 87545

DISTRIBUTION LIST

No. of Copies	Organization
1	Headquarters National Aeronautics and Space Administration Washington, DC 20546
1	Director National Aeronautics and Space Administration Langley Research Center Hampton, VA 23365
1	Director National Aeronautics and Space Administration Manned Spacecraft Center ATTN: Lib Houston, TX 77058
1	AFELM, The Rand Corp. ATTN: Library-D 1700 Main Street Santa Monica, CA 90406
1	BLM Applied Mechanics Consultants ATTN: Dr. A. Boresi 3310 Willett Drive Laramie, WY 82070
1	CALSPAN Corp. ATTN: E. Fisher P.O. Box 400 Buffalo, NY 14225
1	S&D Dynamics, Inc. ATTN: Dr. M. Soifer 755 New York Avenue Huntington, NY 11743
1	Southwest Research Institute ATTN: P. Cox 8500 Culebra Road San Antonio, TX 78228
1	Stanford University Stanford Linear Accelerator Center ALAC, P.O. Box 4349 Stanford, CA 94305

No. of Copies	Organization
1	State University of New York Department of Mathematics ATTN: Dr. Ram Shrivastav Stony Brook, NY 11794
2	University of Delaware Department of Mathematics Department of Mechanical Engr. Newark, DE 19711
1	Princeton University Department of Civil Engineering ATTN: Dr. A.C. Eringen Princeton, NJ 08540
1	University of Kentucky Department of Computer Science ATTN: Prof. H.C. Thacher, Jr. 915 Patterson Office Tower Lexington, KY 40506
1	University of Maryland Institute for Physical Science & Technology ATTN: Dr. Frank Olver College Park, MD 20742
2	University of Wisconsin-Madison Mathematics Research Center ATTN: Dr. John Nohel Dr. Richard Meyer 610 Walnut Street Madison, WI 53706
1	Virginia Commonwealth University Department of Math. Sciences 901 W. Franklin Richmond, VA 23284
1	West Virginia University ATTN: Prof. G. Inger Dept. of Mech & Aerospace Eng. Morgantown, W VA 26506

DISTRIBUTION LIST

No. of Copies	Organization
	Aberdeen Proving Ground
	Dir, USAMSAA
	ATTN: DRXSY-D
	DRXSY-MP, H. Cohen
	Dir, MTD
	ATTN: H. King
	P. Paules
	Cdr, USATECOM
	ATTN: DRSTE-TO-F
	Dir, USACSL, EA, Bldg E-3516
	ATTN: DRDAR-CLB-PA
	DRDAR-CLN
	DRDAR-CLJ-L

USER EVALUATION OF REPORT

Please take a few minutes to answer the questions below; tear out this sheet, fold as indicated, staple or tape closed, and place in the mail. Your comments will provide us with information for improving future reports.

1. BRL Report Number ____________________________

2. Does this report satisfy a need? (Comment on purpose, related project, or other area of interest for which report will be used.)

____

____

____

3. How, specifically, is the report being used? (Information source, design data or procedure, management procedure, source of ideas, etc.) ____________________________

____

____

4. Has the information in this report led to any quantitative savings as far as man-hours/contract dollars saved, operating costs avoided, efficiencies achieved, etc.? If so, please elaborate.

____

____

5. General Comments (Indicate what you think should be changed to make this report and future reports of this type more responsive to your needs, more usable, improve readability, etc.) __________

____

____

____

6. If you would like to be contacted by the personnel who prepared this report to raise specific questions or discuss the topic, please fill in the following information.

Name: ____________________________

Telephone Number: ____________________________

Organization Address: ____________________________

____

____

FOLD HERE

Director
US Army Ballistic Research Laboratory
ATTN: DRDAR-BLA-S
Aberdeen Proving Ground, MD 21005

NO POSTAGE NECESSARY IF MAILED IN THE UNITED STATES

OFFICIAL BUSINESS
PENALTY FOR PRIVATE USE, $300

BUSINESS REPLY MAIL
FIRST CLASS PERMIT NO 12062 WASHINGTON, DC

POSTAGE WILL BE PAID BY DEPARTMENT OF THE ARMY

Director
US Army Ballistic Research Laboratory
ATTN: DRDAR-BLA-S
Aberdeen Proving Ground, MD 21005

FOLD HERE